
1 Repeated Measures

Any measurement that can be repeated (either across time or across space) can be analyzed under this broad
heading. Crowder and Hand[2] describe repeated measures as the situation in which measurements “are made
of the same characteristic on the same observational unit but on more than one occasion.” This is what is
meant by the term longitudinal data. The scope of repeated measures can be expanded to include clustered

data as well; that is, measurements on members of a cluster that are related in some way. For longitudinal
data, a common occurance is for measurements on the same observational unit to be correlated; for cluster
data, the same phenomenon can occur, with clusters playing the role of observational units, and repetition
of measurement occuring within a cluster. In both cases, the usual model assumption of independent errors
may be violated, so a model that can incorporate this lack of independence is needed.

2 Covariance Structure

We assume N observational units (individuals or clusters) and ni observations of the response for the ith unit,
i = 1, . . . , N (if there are no missing values, then ni = n). The observations for the ith unit are coded in
the vector yi, which has length ni. The design matrices Xi and Zi consist of q columns and ni rows. The
entries of Xi are the values taken by the q continuous explanatory variables and/or design indicators of fixed
effects across the ni measurements of the ith unit; if any between-subjects random effects exist, their design
indicators appear in Zi. For the model

yi = Xiβ + Ziγi + εi
[

γi

εi

]

∼ N

(

0,

[

Gi 0
0 Ri

])

the term Covarianace Structure is used to describe how the matrices Gi and Ri are constrained in the (Normal
case of the) general linear mixed model:

y1, . . . ,yn
ind
∼ Nni

(Xiβ, Vi)

where Vi = ZiGiZ
T
i + Ri is called the Variance-Covariance Matrix of the ith unit. It is important to note

that the decomposition of Vi into the Gi term and the Ri term is not necessarily unique; that is, there may
be two or more structures for Gi and Ri that yield the same Vi. An example of this phenomenon will be
given later. Typically, it is assumed that all of the Vis take the same form[3]. The examples that follow are
forms of Ri, but can be used for Gi as well:

2.1 Unstructured

This is the most general form:

Ri = σ2

ρ21 1 ρ2ni

...











In this case the varaince-covariance matrix contains ni (ni − 1) /2 + 1 nuisance parameters to be estimated,
so in practice, estimation of this structure may only convergence for N � ni. Likewise the statistical power
under this structure is reduced since the only “constraint” on Ri is that it be symmetric.



2.2 ARMA(1,1)

The more structured forms of Vi involve assuming that some or all of the ρjks are a function of the “distance”
between observations j and k. If measurements are repeated across time, then the time variable will be
prominent in this distance function. If measurements are repeated across location, then the distance function
will involve some spatial metric reflecting the experimental design’s geometry. One of the more common
structures is the first order autoregressive, first order moving-average model, abbreviated ARMA(1,1):

Ri = σ2





Notice that now there are only three parameters to be estimated, σ2, γ, and ρ. A special case of ARMA(1,1)
occurs when γ = ρ; this is called ARMA(1,0) or just simply AR(1), and only two covariance parameters σ2

and ρ are estimated.

2.3 Equicorrelation

Another model assumes all repeated measurements are equally correlated:

Ri = σ2

This structure (also called spherical or exchangeable) might be applicable to cluster data, where ρ is called the
intra-class correlation coefficient between two members of the same cluster and is “a relative measure of the
within-cluster similarity.”[3] A special case of equicorrelation, called compound symmetry, arises by enforcing

ρ = σ2
a

σ2
a+σ2

e

for some σ2
a and σ2

e . In that case, if σ2 = σ2
a + σ2

e , then

Ri =

2.4 Uncorrelated

Finally, we compare to the fixed effects case where independence is assumed over repeated measurements:

Ri = σ2

1 0 · · · 0
0 1 0



This helps to show (when Gi = 0) that the standard linear model can be thought of as a special case of
repeated measures.



3 Software Packages

3.1 Repeated Measures in SAS

Both PROC GLM and PROC MIXED provide a REPEATED statement by which one or more repeated measures can
be specified. However, there are many nontrivial differences between these two procedures. Some of the more
important differences are:

� PROC GLM has as a model assumption that Ri is either unstructured or of “Type H” structure[5], whereas
PROC MIXED allows a choice between over 30 different structures for Ri through the TYPE= option of the
REPEATED statement.

� PROC GLM ignores observational units with missing observations, whereas PROC MIXED includes them.
(Though note that they must be missing at random for the estimators to remain unbiased.)

� PROC GLM assumes covariates are constant within observational units, whereas in PROC MIXED they are
allowed to vary.

See[6] for a longer treatment of the differences. Note that if REPEATED is omitted, SAS uses the uncorrelated
Ri from section (2.4).

3.1.1 Unstructued

Seven subjects had their response times measured when a light was flashed into each eye through lenses of
powers 6/6, 6/18, 6/36, and 6/60 (a lens of power a/b means that the eye will perceive as being at a feet an
object actually positioned at b feet). Measurements were made in milliseconds, and the question of interest
was whether response time varied with lens strength.[4] Note that there are not enough subjects to fit an
unstructured model for both left and right eyes.

Data Vision; /* From Everitt 1996:103 */

Input Subject @;

Do Eye="Left ","Right";

Do LensStrength="6/06","6/18","6/36","6/60";

Input ResponseTime @;

Output;

End;

End;

DataLines;

1 116 119 116 124 120 117 114 122

2 110 110 114 115 106 112 110 110

3 117 118 120 120 120 120 120 124

4 112 116 115 113 115 116 116 119

5 113 114 114 118 114 117 116 112

6 114 115 94 116 100 99 94 97

7 110 110 105 118 105 105 115 115

;

run;

Proc Mixed Data=Vision;



Where Eye="Left ";

Class Subject LensStrength;

Model ResponseTime = LensStrength;

Repeated LensStrength / Subject=Subject Type=UN R RCorr;

LSMeans LensStrength / PDiff;

Run;

3.1.2 AR(1)

The milk production of four cows is followed over four lactation periods and four diets; assume that diet does
not have any residual effect.[7]

data new;

input cow period trt resp @@;

cards;

1 1 1 38 1 2 2 32 1 3 3 35 1 4 4 33

2 1 2 39 2 2 3 37 2 3 4 36 2 4 1 30

3 1 3 45 3 2 4 38 3 3 1 37 3 4 2 35

4 1 4 41 4 2 1 30 4 3 2 32 4 4 3 33

;

run;

proc mixed data = new;

class cow trt period;

model resp = trt period;

random cow;

repeated period / type = ar(1) subject = cow;

run;

3.1.3 Compound Symmetry

Over the course of a month, three different methods of throwing a baseball were taught to 21 subjects with
seven subjects per method. The throwing speed for each subject was recorded at two and four weeks and
adjusted for their initial throwing speed.[1] Note that subjects are nested within method:

data new;

input meth subj time1 time2;

cards;

1 1 25.4 30.6

1 2 27.4 29.3

1 3 25.5 30.0

1 4 25.8 29.7

1 5 26.2 31.3

1 6 24.6 26.6

1 7 25.6 28.0

2 1 27.6 27.1

2 2 24.7 29.0

2 3 26.3 27.3



2 4 25.0 29.7

2 5 25.7 29.5

2 6 28.5 29.7

2 7 22.9 27.2

3 1 22.8 25.1

3 2 24.2 24.0

3 3 25.3 25.2

3 4 25.4 24.7

3 5 24.5 26.2

3 6 25.6 26.9

3 7 25.6 24.8

;

run;

data new1;

set new;

resp=time1; time=1; output;

resp=time2; time=2; output;

run ;

proc mixed data=new1;

class meth subj time;

model resp= meth|time;

repeated / type = cs sub = subj;

run;

3.2 Repeated Measures in SPSS

(To be completed)

